<commit_before># Ported from CMakeFindDependencyMacro.cmake (finding configs and using pkgconfig as fallback)
set(cmake_quiet_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_QUIETLY)
  set(cmake_quiet_arg QUIET)
endif()
set(cmake_required_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_REQUIRED)
  set(cmake_required_arg REQUIRED)
endif()

find_package(Ogg CONFIG ${cmake_quiet_arg})
if(NOT TARGET Ogg::ogg)
  find_package(PkgConfig REQUIRED ${cmake_quiet_arg})
  pkg_check_modules(Ogg ${cmake_required_arg} ${cmake_quiet_arg} IMPORTED_TARGET ogg)
  add_library(Ogg::ogg ALIAS PkgConfig::Ogg)
endif()

if (NOT TARGET Ogg::ogg)
  set(${CMAKE_FIND_PACKAGE_NAME}_NOT_FOUND_MESSAGE "${CMAKE_FIND_PACKAGE_NAME} could not be found because dependency Ogg could not be found.")
  set(${CMAKE_FIND_PACKAGE_NAME}_FOUND False)
  return()
endif()

find_package(Opus CONFIG ${cmake_quiet_arg})
if(NOT TARGET Opus::opus)
  find_package(PkgConfig REQUIRED ${cmake_quiet_arg})
  pkg_check_modules(Opus ${cmake_required_arg} ${cmake_quiet_arg} IMPORTED_TARGET opus)
  add_library(Opus::opus ALIAS PkgConfig::Opus)
endif()<commit_msg>Make pkgconfig targets globally imported targets<commit_after># Ported from CMakeFindDependencyMacro.cmake (finding configs and using pkgconfig as fallback)
set(cmake_quiet_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_QUIETLY)
  set(cmake_quiet_arg QUIET)
endif()
set(cmake_required_arg)
if(${CMAKE_FIND_PACKAGE_NAME}_FIND_REQUIRED)
  set(cmake_required_arg REQUIRED)
endif()

find_package(Ogg CONFIG ${cmake_quiet_arg})
if(NOT TARGET Ogg::ogg)
  find_package(PkgConfig REQUIRED ${cmake_quiet_arg})
  pkg_check_modules(Ogg ${cmake_required_arg} ${cmake_quiet_arg} IMPORTED_TARGET ogg)
  set_target_properties(PkgConfig::Ogg PROPERTIES IMPORTED_GLOBAL TRUE)
  add_library(Ogg::ogg ALIAS PkgConfig::Ogg)
endif()

if (NOT TARGET Ogg::ogg)
  set(${CMAKE_FIND_PACKAGE_NAME}_NOT_FOUND_MESSAGE "${CMAKE_FIND_PACKAGE_NAME} could not be found because dependency Ogg could not be found.")
  set(${CMAKE_FIND_PACKAGE_NAME}_FOUND False)
  return()
endif()

find_package(Opus CONFIG ${cmake_quiet_arg})
if(NOT TARGET Opus::opus)
  find_package(PkgConfig REQUIRED ${cmake_quiet_arg})
  pkg_check_modules(Opus ${cmake_required_arg} ${cmake_quiet_arg} IMPORTED_TARGET opus)
  set_target_properties(PkgConfig::Opus PROPERTIES IMPORTED_GLOBAL TRUE)
  add_library(Opus::opus ALIAS PkgConfig::Opus)
endif()
{'fraction_non_alphanumeric': 0.0982367758186398, 'fraction_numerical': 0.0, 'mean_word_length': 9.544247787610619, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '470903', 'n_tokens_mistral': 955, 'n_tokens_neox': 912, 'n_words': 168}